Soluzioni del compito di Metodi Matematici e Calcolo per Ingegneria del 1 febbraio 2005

Enrico Bertolazzi

Trasformata di Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente funzione

```
> f := t -> 1-t+t^3 ;
```

$$f := t \to 1 - t + t^3$$

Usando le regole di trasformazione calcolare le trasformate delle funzioni

```
> f1 := t -> f(t)*exp(-2*t) ;
  f2 := t -> f(2*t) ;
  f3 := t -> D(D(f))(t) ;
```

$$f1 := t \to f(t)\, \mathbf{e}^{(-2\, t)}$$

$$f2 := t \to f(2\, t)$$

$$f3 := t \to \mathrm{D}(\mathrm{D}(f))(t)$$

Trasformate con le primitive Maple

```
> laplace(f(t),t,s);
  laplace(f1(t),t,s);
  laplace(f2(t),t,s);
  laplace(f3(t),t,s);
```

$$\frac{1}{s} - \frac{1}{s^2} + \frac{6}{s^4}$$

$$\frac{s^3 + 5\, s^2 + 8\, s + 10}{(s+2)^4}$$

$$\frac{1}{s} - \frac{2}{s^2} + \frac{48}{s^4}$$

$$\frac{6}{s^2}$$

Soluzione di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente equazione differenziale

```
> ode := diff(y(x),x,x)-y(x)=sin(-x) ;
```

$$ode := \left(\frac{d^2}{dx^2}\,y(x)\right) - y(x) = -\sin(x)$$

Con dato iniziale

```
> y0, yp0 := 1,2 ;
```

$$y0, yp0 := 1, 2$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo la equazione differenziale con la trasformata di Laplace

```
> sode := laplace(ode,x,s) ;
```

$$sode := s^2\, laplace(y(x), x, s) - \mathrm{D}(y)(0) - s\, y(0) - laplace(y(x), x, s) = -\frac{1}{s^2+1}$$

Risolvo la equazione per y(s)

```
> lode := isolate(sode,laplace(y(x),x,s));
```

$$lode := laplace(y(x), x, s) = \frac{-\frac{1}{s^2+1} + \mathrm{D}(y)(0) + s\, y(0)}{s^2 - 1}$$

Applico le condizioni iniziali ottenendo y(s)

```
> ly := subs( y(0)=y0,D(y)(0)=yp0,rhs(lode)) ;
```

$$ly := \frac{-\frac{1}{s^2+1} + 2 + s}{s^2 - 1}$$

Espansione in fratti semplici

```
> convert(ly, parfrac, s);
```

$$\frac{1}{2\,(s^2+1)} + \frac{5}{4\,(s-1)} - \frac{1}{4\,(s+1)}$$

Antitrasformo per ottenere la equazione y(x)

```
> res := invlaplace(ly,s,t) ;
```

$$res := -\frac{1}{4}\,\mathbf{e}^{(-t)} + \frac{5}{4}\,\mathbf{e}^{t} + \frac{1}{2}\sin(t)$$

Soluzione di un sistema di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Dato il seguente sistema di equazioni differenziali

```
> ode1 := diff(y(x),x)-z(x)=0 ;
  ode2 := diff(y(x),x)+diff(z(x),x)=1 ;
```

$$ode1 := \left(\frac{d}{dx}\, y(x)\right) - z(x) = 0$$

$$ode2 := \left(\frac{d}{dx}\, y(x)\right) + \left(\frac{d}{dx}\, z(x)\right) = 1$$

Con dato iniziale

```
> y0, z0 := 1,2 ;
```

$$y0, z0 := 1, 2$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo le equazioni differenziale con la trasformata di Laplace

```
> sode1 := laplace(ode1,x,s) ;
  sode2 := laplace(ode2,x,s) ;
```

$$sode1 := s\, laplace(y(x), x, s) - y(0) - laplace(z(x), x, s) = 0$$

$$sode2 := s\, laplace(y(x), x, s) - y(0) + s\, laplace(z(x), x, s) - z(0) = \frac{1}{s}$$

Risolvo la equazione per y(s), z(s)

```
> RES := solve({sode1,sode2},{laplace(y(x),x,s),laplace(z(x),x,s)});
```

$$RES := \left\{ laplace(y(x), x, s) = \frac{1 + z(0)\, s + y(0)\, s^2 + y(0)\, s}{s^2\, (s+1)},\ laplace(z(x), x, s) = \frac{1 + z(0)\, s}{s\, (s+1)} \right\}$$

Applico le condizioni iniziali ottenendo y(s), z(s)

```
> SOL :=
  subs(RES,y(0)=y0,z(0)=z0,<laplace(y(x),x,s),laplace(z(x),x,s)>);
```

$$SOL := \begin{bmatrix} \dfrac{1 + 3\, s + s^2}{s^2\, (s+1)} \\ \\ \dfrac{1 + 2\, s}{s\, (s+1)} \end{bmatrix}$$

Antitrasformo per ottenere y(x), z(x)

```
> yy := invlaplace(SOL[1],s,x) ;
  zz := invlaplace(SOL[2],s,x) ;
```

$$yy := 2 + x - \mathbf{e}^{(-x)}$$

$$zz := 1 + \mathbf{e}^{(-x)}$$

Espansione in fratti semplici per controllo

```
> convert(SOL[1], parfrac, s);
  convert(SOL[2], parfrac, s);
```

$$\frac{2}{s} - \frac{1}{s+1} + \frac{1}{s^2}$$

$$\frac{1}{s}+\frac{1}{s+1}$$

Soluzione di ricorrenza con trasformata zeta

```
> restart:
```

Risolvere la seguente ricorrenza

```
> RIC := f(n+2) = 2*f(n+1) - f(n) + 1;
```

$$RIC := f(n+2) = 2\,f(n+1) - f(n) + 1$$

Con dato iniziale

```
> INI := f(0)=0,f(1)=1 ;
```

$$INI := f(0) = 0, f(1) = 1$$

Usando le primitive si maple:

```
> rsolve({RIC,INI}, f(k));
```

$$-k - 1 + (k+1)\left(\frac{1}{2}\,k + 1\right)$$

```
> simplify(%);
```

$$\frac{1}{2}\,k + \frac{1}{2}\,k^2$$

Usando la Z-trasformata (occhio negli appunti abbiamo Z(f) = sum(f(n)*z^n mentre maple usa Z(f) = sum(f(n)*w^{-n}
Quindi per confrontare le trasformate bisogna trasfromare w -> 1/z;

```
> zRIC := ztrans(RIC,n,w):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRIC)) ;
```

$$\frac{f(z) - f(0) - f(1)\,z}{z^2} = -\frac{2\,f(z) - 3\,f(z)\,z - 2\,f(0) + 2\,f(0)\,z + f(z)\,z^2 + z}{z\,(-1+z)}$$

Ricavo f(w) [f(1/z)]

```
> zRICrhs := isolate(zRIC,ztrans(f(n),n,w)):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhs)) ;
```

$$f(z) = -\frac{f(0) - 3\,f(0)\,z + f(1)\,z - f(1)\,z^2 + 2\,f(0)\,z^2 + z^2}{(-1+z)\,(1 - 2\,z + z^2)}$$

Applico le condizioni iniziali

```
> zRICrhsINI := subs(INI,zRICrhs):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhsINI)) ;
```

$$f(z) = -\frac{z}{(-1+z)\,(1 - 2\,z + z^2)}$$

Conversione in fratti semplici

```
> convert(%, parfrac);
```

$$f(z) = -\frac{1}{(-1+z)^3} - \frac{1}{(-1+z)^2}$$

Inversione della Z-trasformata

```
> invztrans(zRICrhsINI,w,k) ;
```

$$f(k) = \frac{1}{2}k + \frac{1}{2}k^2$$

Soluzione di un sistema non lineare con Newton

```
> restart:
  with(VectorCalculus):
Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected:
`*`, `+`, `.`, D, Vector, diff, int, limit, series
```

Sistema non lineare

```
> f := exp(x+y)-1 ;
  g := exp(x-y)-1 ;
```

$$f := e^{(x+y)} - 1$$

$$g := e^{(x-y)} - 1$$

Soluzione esatta

```
> solve({f,g},{x,y}) ;
```

$$\{x = 0, y = 0\}$$

Matrice Jacobiano

```
> J := Jacobian([f,g],[x,y]) ;
```

$$J := \begin{bmatrix} e^{(x+y)} & e^{(x+y)} \\ e^{(x-y)} & -e^{(x-y)} \end{bmatrix}$$

Schema di Newton

```
> Newton_update := <x,y>-J^(-1).<f,g> ;
```

$$Newton\_update := \left(x - \frac{e^{(x+y)} - 1}{2\,e^{(x+y)}} - \frac{e^{(x-y)} - 1}{2\,e^{(x-y)}} \right) e_x + \left(y - \frac{e^{(x+y)} - 1}{2\,e^{(x+y)}} + \frac{e^{(x-y)} - 1}{2\,e^{(x-y)}} \right) e_y$$

Schema di Newton per questo sistema non lineare

```
> x[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[1])) ;
  y[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[2])) ;
```

$$x_{k+1} = x_k - 1 + \frac{1}{2}e^{(-x_k - y_k)} + \frac{1}{2}e^{(-x_k + y_k)}$$

$$y_{k+1} = y_k + \frac{1}{2}\,\mathbf{e}^{(-x_k - y_k)} - \frac{1}{2}\,\mathbf{e}^{(-x_k + y_k)}$$

Tre iterate a partire da (1,2)

```
> x[0],y[0]:= 1,2 ;
```

$$x_0, y_0 := 1, 2$$

Prima iterata

```
> x[1] := evalf(subs(x=x[0],y=y[0],Newton_update[1])) ;
  y[1] := evalf(subs(x=x[0],y=y[0],Newton_update[2])) ;
```

$$x_1 := 1.384034448$$

$$y_1 := 0.6657526200$$

Seconda iterata

```
> x[2] := evalf(subs(x=x[1],y=y[1],Newton_update[1])) ;
  y[2] := evalf(subs(x=x[1],y=y[1],Newton_update[2])) ;
```

$$x_2 := 0.6922102565$$

$$y_2 := 0.4863391297$$

Terza iterata

```
> x[3] := evalf(subs(x=x[2],y=y[2],Newton_update[1])) ;
  y[3] := evalf(subs(x=x[2],y=y[2],Newton_update[2])) ;
```

$$x_3 := 0.2530416410$$

$$y_3 := 0.2332325500$$

Problema di Minimo Vincolato

```
> restart:
  with(LinearAlgebra):
  with(Optimization):
  with(VectorCalculus):
Warning, the names `&x`, CrossProduct and DotProduct have been rebound

Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected:
`*`, `+`, `.`, D, Vector, diff, int, limit, series
```

Minimizzare la seguente funzione

```
> f := z-x-y;
```

$$f := z - x - y$$

Soggetta ai seguenti vincoli

```
> v := [x+y+z^2=3,x^2-z^2=0] ;
```

$$v := [x + y + z^2 = 3, x^2 - z^2 = 0]$$

Soluzione con le primitive Maple

```
> Minimize(f, v );
```
$$[-3.25000000000000044, [z = -0.500000000000018984, x = -0.500000000000018984, y = 3.25000000000000044]]$$

Uso dei moltiplicatori di Lagrange

```
> v1 := lhs(v[1])-rhs(v[1]) ;
  v2 := lhs(v[2])-rhs(v[2]) ;
```
$$v1 := x + y + z^2 - 3$$
$$v2 := x^2 - z^2$$

```
> g := f - lambda*v1 - mu*v2 ;
```
$$g := z - x - y - \lambda\,(x + y + z^2 - 3) - \mu\,(x^2 - z^2)$$

Sistema non lineare da risolvere

```
> F := Gradient(g,[x,y,z,lambda,mu]) ;
```
$$F := (-1 - \lambda - 2\,\mu\,x)\,\overline{e}_x + (-1 - \lambda)\,\overline{e}_y + (1 - 2\,\lambda\,z + 2\,\mu\,z)\,\overline{e}_z + (-x - y - z^2 + 3)\,\overline{e}_\lambda + (-x^2 + z^2)\,\overline{e}_\mu$$

```
> RES := solve({F[1],F[2],F[3],F[4],F[5]},{x,y,z,lambda,mu}) ;
```
$$RES := \left\{\mu = 0, \lambda = -1, y = \frac{9}{4}, x = \frac{1}{2}, z = \frac{-1}{2}\right\}, \left\{\mu = 0, \lambda = -1, y = \frac{13}{4}, x = \frac{-1}{2}, z = \frac{-1}{2}\right\}$$

Controllo proprietà di minimo

```
> Hf  := Hessian(f,[x,y,z]):
  Hv1 := Hessian(v1,[x,y,z]):
  Hv2 := Hessian(v2,[x,y,z]):
  Hf, Hv1, Hv2 ;
```
$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 2 \end{bmatrix}, \begin{bmatrix} 2 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & -2 \end{bmatrix}$$

```
> JH := Jacobian([v1,v2],[x,y,z]) ;
  NH := NullSpace(JH) ;
```
$$JH := \begin{bmatrix} 1 & 1 & 2\,z \\ 2\,x & 0 & -2\,z \end{bmatrix}$$
$$NH := \left\{\begin{bmatrix} \dfrac{z}{x} \\ -\dfrac{z\,(1 + 2\,x)}{x} \\ 1 \end{bmatrix}\right\}$$

Controllo minimo/massimo locale primo punto

```
> lambda1 := subs(RES[1],lambda);
```

```
mu1      := subs(RES[1],mu);
```

$$\lambda 1 := -1$$

$$\mu 1 := 0$$

```
> Hf1 := simplify(Hf - lambda1. Hv1 - mu1. Hv2) ;
```

$$Hf1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES[1],op(NH)) ;
```

$$Z1 := \begin{bmatrix} -1 \\ 2 \\ 1 \end{bmatrix}$$

E' positivo per ogni alpha > 0, quindi è un minimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$2\,\alpha^2$$

Controllo minimo/massimo locale secondo punto

```
> lambda2 := subs(RES[2],lambda);
  mu2      := subs(RES[2],mu);
```

$$\lambda 2 := -1$$

$$\mu 2 := 0$$

```
> Hf2 := simplify(Hf - lambda2. Hv1 - mu2. Hv2) ;
```

$$Hf2 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z2 := subs(RES[2],op(NH)) ;
```

$$Z2 := \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix}$$

```
> simplify(Transpose(alpha.Z2).Hf2.(alpha.Z2)) ;
```

$$2\,\alpha^2$$

E' positivo per ogni alpha > 0, quindi è un minimo locale.

Calcolo i valori:

```
> subs(RES[1],f);
```

$$-\frac{13}{4}$$

```
> subs(RES[2],f);
```

$$-\frac{13}{4}$$

```
>
```

Sono entrami mimimi assoluti.